

 THE POWER TO KNOW.




Have It Your Way: Rearrange and Replay Your Output with ODS Document

THE POWER TO KNOW.

Cynthia Zender
SAS Global Forum 2009

Copyright © 2008, SAS Institute Inc. All rights reserved.

 THE POWER TO KNOW.

Today's Topics

- Overview of ODS Document
- Uses for ODS Document
- Creating, Managing and Replaying an ODS Document
- Using Proc Document


This paper (and programs) will be available for download at:

<http://support.sas.com/rnd/papers>
(search by title, not author)

Copyright © 2008, SAS Institute Inc. All rights reserved.

What is an ODS Document?

- ODS DOCUMENT is an ODS destination.
- ODS DOCUMENT is a "data" destination, not a report destination. Output objects are stored in their "raw" form, as if they were frozen between procedure completion and ODS result file creation.



Copyright © 2008, SAS Institute Inc. All rights reserved.


What is an ODS Document?

- ODS DOCUMENT destination
 - holds output objects that are based on a SAS procedure or process in their original creation hierarchy
 - allows management and restructuring of output objects by GUI interface or batch syntax
 - allows ODS documents to be replayed to ODS report destinations.

Copyright © 2008, SAS Institute Inc. All rights reserved.

What is an ODS Document?

- ODS documents are stored in a proprietary format (a document store) and can be viewed only with SAS software.
- To view or modify what's in the document store, you have to use either the ODS DOCUMENT window or PROC DOCUMENT.



What is PROC DOCUMENT?

- PROC DOCUMENT is an interactive procedure that provides batch syntax to
 - create, manage, rearrange and replay ODS Document stores
 - open and list objects in a document store
 - display or replay output to any open ODS report destination without rerunning the SAS program(s) that created the output objects.

Creating a Document Store

```
ods document name=work.prddoc(write);
proc tabulate data=prdsale;
  by Country;
  var predict;
  class prodtype;
  table prodtype all,
        predict*(min mean max);
run;

ods select ExtremeObs;
proc univariate data=prdsale;
  by Country;
  var actual;
run;
ods document close;
```

demo1_make_odsdoc.sas

Copyright © 2008, SAS Institute Inc. All rights reserved.

Results Window

The screenshot shows the SAS Results window. On the left is a tree view with the following structure:

- Results
 - Tabulate
 - Country=CANADA
 - Cross-tabular summary report
 - Table 1
 - Country=GERMANY
 - Cross-tabular summary report
 - Table 1

On the right is the 'Properties of Table:' window, which contains the following table:

| Attribute | Value |
|---------------|---|
| Type | Document |
| Name | Table |
| Description | Table 1 |
| Path | Tabulate#1.ByGroup1#1.Report#1.Table#1 |
| Created | 3/20/2009 12:15 AM |
| Document | Work.Prddoc |
| Document Path | \Tabulate#1\ByGroup1#1\Report#1\Table#1 |

Although you see a document icon in the SAS Results Window after creating the document store, you must use either PROC DOCUMENT or the Document Window to view the objects in the document store.

demo1_make_odsdoc.sas

Copyright © 2008, SAS Institute Inc. All rights reserved.

Batch Syntax to View the Document Store

```
ods listing;

proc document name=work.prddoc;
  list / levels=all;
run;
quit;
```

demo2_investigate_replay.sas

Copyright © 2008, SAS Institute Inc. All rights reserved.

LIST Statement Results

| Listing of: \Work.Prddoc\ Order by: Insertion Number of Levels: All | | |
|---|--|-------|
| Obs | Path | Type |
| 1 | \Tabulate#1 | Dir |
| 2 | \Tabulate#1\ByGroup1#1 | Dir |
| 3 | \Tabulate#1\ByGroup1#1\Report#1 | Dir |
| 4 | \Tabulate#1\ByGroup1#1\Report#1\Table#1 | Table |
| 5 | \Tabulate#1\ByGroup2#1 | Dir |
| 6 | \Tabulate#1\ByGroup2#1\Report#1 | Dir |
| 7 | \Tabulate#1\ByGroup2#1\Report#1\Table#1 | Table |
| 8 | \Tabulate#1\ByGroup3#1 | Dir |
| 9 | \Tabulate#1\ByGroup3#1\Report#1 | Dir |
| 10 | \Tabulate#1\ByGroup3#1\Report#1\Table#1 | Table |
| 11 | \Univariate#1 | Dir |
| 12 | \Univariate#1\ByGroup1#1 | Dir |
| 13 | \Univariate#1\ByGroup1#1\ACTUAL#1 | Dir |
| 14 | \Univariate#1\ByGroup1#1\ACTUAL#1\ExtremeObs#1 | Table |
| 15 | \Univariate#1\ByGroup2#1 | Dir |
| 16 | \Univariate#1\ByGroup2#1\ACTUAL#1 | Dir |
| 17 | \Univariate#1\ByGroup2#1\ACTUAL#1\ExtremeObs#1 | Table |
| 18 | \Univariate#1\ByGroup3#1 | Dir |
| 19 | \Univariate#1\ByGroup3#1\ACTUAL#1 | Dir |
| 20 | \Univariate#1\ByGroup3#1\ACTUAL#1\ExtremeObs#1 | Table |

demo2_investigate_replay.sas

Copyright © 2008, SAS Institute Inc. All rights reserved.

Replaying All Output Objects

```
ods html file='c:\temp\replay_all.html'
    style=sasweb;
ods pdf file='c:\temp\replay_all.pdf';
proc document name=work.prddoc;
    replay;
run;
quit;
ods _all_ close;
```

Copyright © 2008, SAS Institute Inc. All rights reserved.

REPLAY Statement Results

The screenshot shows the SAS interface with the 'REPLAY Statement Results'. The left pane contains a tree view of the document structure. The main pane displays a preview of the output for 'Country=CANADA', showing a table with 'Predicted Value' and 'Observed Value' for 'ACTUAL' and 'Extreme Observations'.

| Predicted Value | Observed Value |
|----------------------|----------------|
| ACTUAL | 100.00 |
| Extreme Observations | 100.00 |

The screenshot shows the 'Results' pane in the SAS interface. The tree view on the right lists the output objects, including 'Tabulate: The Tabulate Procedure' for Canada, Germany, and the U.S.A., and 'Univariate: The Univariate Procedure' for the same countries. The main pane shows a preview of the output for 'Country=CANADA', displaying a table with 'Predicted Value' and 'Observed Value' for 'ACTUAL' and 'Extreme Observations'.

| Predicted Value | Observed Value |
|----------------------|----------------|
| ACTUAL | 100.00 |
| Extreme Observations | 100.00 |


Copyright © 2008, SAS Institute Inc. All rights reserved.

Viewing the Document Store Interactively

To open the Document Window, use one of these two methods:
Select **View** → **Documents** from the View menu

OR


type '**odsdocument**' or '**odsdoc**' in the command area.



Copyright © 2008, SAS Institute Inc. All rights reserved.

The ODS Document Window


- The ODS Document window provides a GUI window to explore an ODS Document store.



Copyright © 2008, SAS Institute Inc. All rights reserved.

Docked View


- The ODS Document window, when open, is usually docked with the SAS Explorer window and the SAS Results window.
- The ODS Document window is not available with SAS Enterprise Guide. However, the PROC DOUCMENT batch syntax can be used with Enterprise Guide.



Copyright © 2008, SAS Institute Inc. All rights reserved.

Replay an ODS Document Interactively


To replay an entire Document store, right-click on the store name and select **Open As...**



Copyright © 2008, SAS Institute Inc. All rights reserved.

Select a Destination


- You may choose a destination from the Open As dialogue box. (Choose multiple destinations with Ctrl + Click.)



Copyright © 2008, SAS Institute Inc. All rights reserved.

PDF Results


- The final PDF results show the ODS Document structure, which is the same as the original structure.



Copyright © 2008, SAS Institute Inc. All rights reserved.

Replaying Selected Objects

- You do not have to replay the entire document store, you can also replay selected objects in the document store.



Copyright © 2008, SAS Institute Inc. All rights reserved.

HTML Output for ExtremeObs Object

- The single object was replayed from the document store.


The screenshot shows the 'Documents' window on the left with the same tree structure as the first slide. The 'Open As...' option in the context menu is highlighted. An arrow points from this menu item to the right, where a preview of the HTML output is shown. The output is titled 'The UNIVARIATE Procedure Variable: ACTUAL (Actual Sales)' and 'Country=CANADA'. It contains a table of 'Extreme Observations'.

| Extreme Observations | | | |
|----------------------|-----|---------|-----|
| Lowest | | Highest | |
| Value | Obs | Value | Obs |
| 3 | 377 | 995 | 74 |
| 5 | 121 | 995 | 195 |
| 6 | 296 | 997 | 239 |
| 8 | 127 | 999 | 2 |
| 9 | 299 | 1000 | 455 |

Copyright © 2008, SAS Institute Inc. All rights reserved.

Creating a New Document Store Interactively


- To create a new ODS document in this document store, highlight the word '**Documents**' at the top of the Document window and select **New Document** from the menu.



Copyright © 2008, SAS Institute Inc. All rights reserved.

Providing Document Information


- When the New Document window opens, select a library to store the new document store and provide a name for the document store, NEWPRD, for example. Then, click **OK**.



Copyright © 2008, SAS Institute Inc. All rights reserved.

Creating a New Folder


- To create a folder in the new document store, highlight the name of the new document store, right-click, and select **New Folder**.



Copyright © 2008, SAS Institute Inc. All rights reserved.

Create the New Folder Structure


- Enter the country and the description in the New Folder window and then select OK. Folders for Canada and Germany were already created prior to this screen shot for the creation of the U.S.A. folder. Note that if a Description value is provided, it will be the value used.



Copyright © 2008, SAS Institute Inc. All rights reserved.

New Folders Created


- The folder with all three countries is shown here. Even though the folders are expanded, they have no objects, so there is nothing to show in the expanded view.



Copyright © 2008, SAS Institute Inc. All rights reserved.

New Document Structure


- If your company requires any type of audit trail of output or if you might ever need to reproduce the original report, you should keep an archive copy of the original output structure intact.
- You can either copy and paste or drag and drop from the original document store (WORK.PRDDOC) to the new document store.



Copyright © 2008, SAS Institute Inc. All rights reserved.

Replay New ODS Document


- If you follow the same Open As... steps and replay the document to the PDF destination, the new structure is displayed.



Copyright © 2008, SAS Institute Inc. All rights reserved.

Batch Syntax

- PROC DOCUMENT provides a batch syntax language to create, manage and replay a document store.
- PROC DOCUMENT is a "run-time" procedure. Use RUN to execute groups of statements and use QUIT to terminate the procedure.



Copyright © 2008, SAS Institute Inc. All rights reserved.

The MAKE Statement

- The MAKE statement creates three directories or folders, one for each country. The highlighted code was submitted and the results are immediately shown in the open Documents window.

```


proc document name=work.neworder(write);
  make CANADA, GERMANY, USA;
run;

dir ^^;
dir CANADA;
dir;
copy \work.prddoc\Tabulate#1\ByGroup1#1\Report#1\Table#1 to ^;
copy \work.prddoc\Univariate#1\ByGroup1#1\ACTUAL#1\ExtremeObs#1
    to ^;
run;

```

Navigating the Document Store

- In order to use the batch syntax, you have to navigate the directory structure, using the DIR statement.



The COPY Statement

- Use the COPY statement to copy output objects from one document store to a new document store.
- The navigation "shortcut" **to ^** means "to the current directory in the document store".


```
copy \work.prddoc\Tabulate#1\ByGroup1#1\Report#1\Table#1 to ^;

copy \work.prddoc\Univariate#1\ByGroup1#1\ACTUAL#1\ExtremeObs#1
to ^ ;
```

Copyright © 2008, SAS Institute Inc. All rights reserved.

Using Object Names

- Note how the COPY statement uses the object names, as revealed in the LIST output.



```
copy \work.prddoc\Univariate#1\ByGroup1#1\ACTUAL#1\ExtremeObs#1
to ^ ;

run;

dir ^^;
dir GERMANY;
dir;
copy \work.prddoc\Tabulate#1\ByGroup2#1\Report#1\Table#1 to ^ ;
copy \work.prddoc\Univariate#1\ByGroup2#1\ACTUAL#1\ExtremeObs#1
to ^ ;

run;
```

Listing of: \Work.Prddoc\
Order by: Insertion
Number of levels: All

| Obs | Path | Type |
|-----|--|-------|
| 1 | \Tabulate#1 | Dir |
| 2 | \Tabulate#1\ByGroup1#1 | Dir |
| 3 | \Tabulate#1\ByGroup1#1\Report#1 | Dir |
| 4 | \Tabulate#1\ByGroup1#1\Report#1\Table#1 | Table |
| 5 | \Tabulate#1\ByGroup2#1 | Dir |
| 6 | \Tabulate#1\ByGroup2#1\Report#1 | Dir |
| 7 | \Tabulate#1\ByGroup2#1\Report#1\Table#1 | Table |
| 8 | \Tabulate#1\ByGroup3#1 | Dir |
| 9 | \Tabulate#1\ByGroup3#1\Report#1 | Dir |
| 10 | \Tabulate#1\ByGroup3#1\Report#1\Table#1 | Table |
| 11 | \Univariate#1 | Dir |
| 12 | \Univariate#1\ByGroup1#1 | Dir |
| 13 | \Univariate#1\ByGroup1#1\ACTUAL#1 | Dir |
| 14 | \Univariate#1\ByGroup1#1\ACTUAL#1\ExtremeObs#1 | Table |
| 15 | \Univariate#1\ByGroup2#1 | Dir |
| 16 | \Univariate#1\ByGroup2#1\ACTUAL#1 | Dir |
| 17 | \Univariate#1\ByGroup2#1\ACTUAL#1\ExtremeObs#1 | Table |
| 18 | \Univariate#1\ByGroup3#1 | Dir |
| 19 | \Univariate#1\ByGroup3#1\ACTUAL#1 | Dir |
| 20 | \Univariate#1\ByGroup3#1\ACTUAL#1\ExtremeObs#1 | Table |


Arrows point from the 'dir' commands in the code to the corresponding rows in the listing output.

Copyright © 2008, SAS Institute Inc. All rights reserved.

Batch Replay Syntax

- The REPLAY statement of PROC DOCUMENT provides the equivalent of the Open As... interactive method of replaying documents. This also allows you to specify ODS options to impact the result file and select a name for the result file.

```
ods html file='replay_new.html'
      style=sasweb;
ods pdf file='replay_new.pdf';
proc document name=work.neworder;
  replay;
run;
quit;
ods _all_ close;
```



Copyright © 2008, SAS Institute Inc. All rights reserved.

Special Subsetting Variables

- WHERE expression processing is possible starting in SAS® 9.2 because ODS DOCUMENT now surfaces a special group of subsetting variables.
- The variables work with the following PROC DOCUMENT statements: COPY TO, MOVE TO, LIST, REPLAY, and DELETE.

| Subsetting Variable | Meaning |
|---------------------|--|
| _CDATE_ | creation date of the current entry |
| _LABEL_ | label of the current entry |
| _LABELPATH_ | path to the label of the current entry |
| _MDATE_ | modification date of the current entry |
| _NAME_ | name of the current entry |
| _PATH_ | path of the current entry |
| _TYPE_ | type of the current entry (Table or Graph) |
| Variable-name | name of a BY variable |

Copyright © 2008, SAS Institute Inc. All rights reserved.

Replaying Selected Output Objects

- PROC DOCUMENT supports WHERE processing for selecting output objects for replay.

```

** Replay Using a WHERE expression;
ods pdf file='c:\temp\replay_where.pdf';
proc document name=work.neworder;
    replay ^(where=(_path_ contains 'Table#1')) ;
run;
quit;
ods _all_ close;

** Replay a directory by directory name;
ods pdf file='c:\temp\replay_dir.pdf';
proc document name=work.neworder;
    replay \CANADA#1 / levels=all;
run;
quit;
ods _all_ close;

```

Copyright © 2008, SAS Institute Inc. All rights reserved.

Replaying Selected Output Objects

- PROC DOCUMENT supports WHERE processing based on using a BY variable name.

```

** Replay Using a BY variable name;
ods html file='c:\temp\replay_var.html'
    style=sasweb;
ods pdf file='c:\temp\replay_var.pdf';
proc document name=work.neworder;
    replay ^(where=(Country = 'GERMANY')) ;
run;
quit;
ods _all_ close;

```

Copyright © 2008, SAS Institute Inc. All rights reserved.

Replaying to Multiple Destinations

- Different output objects can be replayed to different destinations.

```

** Replay to Multiple Destinations;
ods html file='c:\temp\replay_dest.html'
    style=sasweb;
ods pdf file='c:\temp\replay_dest.pdf';
ods rtf file='c:\temp\replay_dest.rtf'
    style=journal startpage=no;
proc document name=work.neworder;
    replay \USA#1 / levels=all dest=(html pdf);
run;

    replay ^(where=( _path_ contains 'Table#1'))
        / dest=rtf ;
run;
quit;
ods _all_ close;

```

Copyright © 2008, SAS Institute Inc. All rights reserved.

Results to Multiple Destinations

Country=CANADA

| Predicted Sales | | |
|-----------------|------|--------|
| | Min | Mean |
| Product type | | |
| FURNITURE | 0.00 | 495.13 |
| OFFICE | 5.00 | 495.23 |
| All | 0.00 | 495.46 |

Country=GERMANY

| Predicted Sales | | |
|-----------------|-------|--------|
| | Min | Mean |
| Product type | | |
| FURNITURE | 0.00 | 495.21 |
| OFFICE | 10.00 | 495.00 |
| All | 0.00 | 497.46 |

Country=U.S.A.

| Predicted Sales | | |
|-----------------|-------|--------|
| | Min | Mean |
| Product type | | |
| FURNITURE | 1.00 | 495.13 |
| OFFICE | 19.00 | 509.23 |
| All | 1.00 | 503.59 |

RTF Output shown in Word Print Preview Mode

Country=U.S.A.

| Predicted Sales | | | |
|-----------------|-------|--------|---------|
| | Min | Mean | Max |
| Product type | | | |
| FURNITURE | 1.00 | 495.13 | 1000.00 |
| OFFICE | 19.00 | 509.23 | 999.00 |
| All | 1.00 | 503.59 | 1000.00 |

The UNIVARIATE Procedure
Variable: ACTUAL (Actual Sales)

Country=U.S.A.

| Extreme Observations | | | |
|----------------------|------|---------|------|
| Lowest | | Highest | |
| Value | Obs | Value | Obs |
| 4 | 961 | 974 | 1366 |
| 6 | 1300 | 977 | 1179 |

HTML Output shown in Internet Explorer

Copyright © 2008, SAS Institute Inc. All rights reserved.

Tasks That Cannot Be Performed Interactively

- There are some actions that can only be accomplished with PROC DOCUMENT syntax. These tasks generally enhance the document and cannot be performed interactively.

| Tasks That You Can Only Perform with PROC DOCUMENT | Documents Window | Document Procedure |
|---|------------------|--------------------|
| Import a data set or graph segment. | No | Yes |
| Set or display the current directory. | No | Yes |
| Create or delete a page break. | No | Yes |
| Create or modify title lines. | No | Yes |
| Create or modify subtitles. | No | Yes |
| Create or modify the lines of text before output objects. | No | Yes |
| Create or modify the lines of text after output objects. | No | Yes |
| Create or modify footnote lines. | No | Yes |
| Create text strings in the current folder. | No | Yes |

Copyright © 2008, SAS Institute Inc. All rights reserved.


Tasks Possible with Both Methods

| Tasks That You Can Perform with Both Methods | Documents Window | Document Procedure |
|---|------------------|--------------------|
| Create a new ODS document. | Yes | Yes |
| Create a new folder. | Yes | Yes |
| Copy folders or output objects. | Yes | Yes |
| Move folders or output objects. | Yes | Yes |
| Create a symbolic link from one output object to another output object. | Yes | Yes |
| Delete a document, folder, or output object. | Yes | Yes |
| Rename a folder or output object. | Yes | Yes |
| Assign a description to a folder or output object. | Yes | Yes |
| Prevent entries from being displayed when they are replayed. | Yes | Yes |
| Show entries that are excluded. | Yes | Yes |
| Enable hidden entries to be displayed. | Yes | Yes |
| Replay to the specified open ODS destinations. | Yes | Yes |
| Determine the path specification. | Yes | Yes |

Copyright © 2008, SAS Institute Inc. All rights reserved.

Desired Enhancements

- Create one page for each country's output by removing the page break between output objects.
- Insert some text specific to each country.
- Create labels for the folders.



Copyright © 2008, SAS Institute Inc. All rights reserved.

OB Statements

- The "OB" statements alter notes, page breaks, titles and footnotes in the ODS document.
- OBPAGE statement allows page breaks to be inserted or deleted.
- OBBNOTE allows the insertion of notes before an output object, while the OBANOTE statement allows the insertion of notes after an output object.
- Other "OB" statements include: OBTITLE, OBFOOTN, OBPAGE, OBSTITLE. Refer to the ODS Document documentation for more information on the use of these statements.

Copyright © 2008, SAS Institute Inc. All rights reserved.

Enhancement Code

- In addition to deleting page breaks and the titles associated with them, different text strings are specified for each country.

```
proc document name=work.neworder(update);
  dir ^^;
  obpage \CANADA#1\ExtremeObs#1 /delete;
  obbnote \CANADA#1\Table#1 'Something' /just=center;
  obbnote \CANADA#1\ExtremeObs#1 'Text1' / just=center;
  obpage \GERMANY#1\ExtremeObs#1 / delete;
  obbnote \GERMANY#1\Table#1 'Something Else' /just=center;
  obbnote \GERMANY#1\ExtremeObs#1 'Text2' / just=center;
  obpage \USA#1\ExtremeObs#1 /delete;
  obbnote \USA#1\Table#1 'Another Thing' /just=center;
  obbnote \USA#1\ExtremeObs#1 'Text3' / just=center;
  setlabel \CANADA#1\Table#1 'Min, Mean, Max';
  setlabel \GERMANY#1\Table#1 'Min, Mean, Max';
  setlabel \USA#1\Table#1 'Min, Mean, Max';
run;
quit;
```

Copyright © 2008, SAS Institute Inc. All rights reserved.

Enhancement Results

```
obbnote \CANADA#1\Table#1 'Something' /just=center;
obbnote \CANADA#1\ExtremeObs#1 'Text1' / just=center;
```

| | | | |
|----------------|------|----------------------|---------|
| Country=CANADA | | | |
| Something | | | |
| | | Predicted Sales | |
| | | Min | Mean |
| Product type | | | Max |
| FURNITURE | 0.00 | 465.44 | 1000.00 |
| OFFICE | 0.00 | 480.00 | 990.00 |
| All | 0.00 | 485.40 | 1000.00 |
| Text1 | | | |
| | | Extreme Observations | |
| | | Lowest | Highest |
| Value | Obs | Value | Obs |
| 3 | 377 | 995 | 74 |
| 5 | 121 | 995 | 195 |
| 6 | 286 | 997 | 239 |
| 8 | 127 | 999 | 2 |
| 9 | 299 | 1000 | 435 |

| | | | |
|-----------------|-------|----------------------|---------|
| Country=GERMANY | | | |
| Something Else | | | |
| | | Predicted Sales | |
| | | Min | Mean |
| Product type | | | Max |
| FURNITURE | 0.00 | 456.01 | 993.00 |
| OFFICE | 10.00 | 450.00 | 992.00 |
| All | 0.00 | 482.40 | 993.00 |
| Text2 | | | |
| | | Extreme Observations | |
| | | Lowest | Highest |
| Value | Obs | Value | Obs |
| 3 | 747 | 994 | 652 |
| 13 | 606 | 996 | 481 |
| 14 | 710 | 996 | 504 |
| 15 | 626 | 996 | 563 |
| 22 | 835 | 1000 | 663 |

```
obbnote \GERMANY#1\Table#1 'Something Else' /just=center;
obbnote \GERMANY#1\ExtremeObs#1 'Text2' / just=center;
```

Copyright © 2008, SAS Institute Inc. All rights reserved.

May the
ODS Document Force
Be With You!

Copyright © 2008, SAS Institute Inc. All rights reserved.



Copyright © 2008, SAS Institute Inc. All rights reserved.

Your Turn



Copyright © 2008, SAS Institute Inc. All rights reserved.

About the Speaker

Speaker Cynthia Zender

Company SAS Institute, Inc.

Telephone (575) 522-3803

Comments & E-Mail Cynthia.Zender@sas.com

This paper and all programs will be available for download at:

<http://support.sas.com/rnd/papers>

(Search by paper title, not by author)

Copyright © 2008, SAS Institute Inc. All rights reserved.

